



RISK DATA OPEN STANDARD

CDL Semantics and Examples

JANUARY 2020

Contents

Contract Definition Language Semantics	
Overview	4
CDL High-Level Flow	5
Order of Operations	5
CDL Covers	6
Attachment	6
Payout	6
Share	6
CDL Terms	7
Sublimits	7
Deductibles	7
Net of Terms	7
CDL Special Topics	8
Amounts	8
Time Basis	8
Sections	8
Cover Subject	9
Other Inputs	9
Detailed Contract Model Examples	10
Simple Covers and Terms	12
Simple Covers	12
Simple Sublimits	13
Simple Deductibles	13
Simple Sections	15

Complex Terms		16
	Max Deductible	16
	Percent of Loss Deductible	16
	Percent of RCV Covered	17
	Percent of RCV Affected	17
	Franchise Deductible	17
	Per Risk Terms	18
	Absorbing a Deductible	18
	Aggregate Terms	19
	Complex Covers	20
	Aggregate Limits and Attachments	20
	Cover Trees	20
	Constrained Covers	21
	Payout Functions	21
	Step Policies	22
	Franchise Attachment	23
Sp	pecial Topics	25
	Expressions	26
	Simple Expressions	26
	Dynamic Expressions	27
	Currency	29
,	Subschedules	31
	Sections	32
	Cashflows	32
	Reinstatements	32

Contract Definition Language Semantics

Overview

Contracts in the Risk Data Open Standard (RDOS) are defined using Contract Definition Language (CDL). CDL is a domain-specific natural language for transparently and unambiguously defining how much a contract pays out given a set of discrete claims. For purposes of CDL semantics, claims are equal to losses from events on dates in a simulation period. CDL applies one period at a time, and the semantics define how CDL handles each dated event within that period.

CDL is not responsible for forming the claims. Insurance contracts typically take losses to exposures as defined by some model. Reinsurance contracts typically take losses to higher-level positions such as portfolios and groups of portfolios. Structure Definition Language (SDL) defines how losses can be combined and processed prior to the application of any reinsurance.

The following topics provide additional information about CDL semantics:

- High-level flow, including order of operations
- Covers—attachment, payout, and share
- Terms—sublimits and deductibles
- Special topics—amounts, time basis, sections, cover subject, and other inputs

Read next: CDL High-Level Flow

CDL High-Level Flow

Losses to a subject comprise the main input to a contract. A subject is defined external to the contract itself, meaning that a contract does not determine its own subject nor how the contract payout is consumed.

Order of Operations

The subject of a contract flows first through any terms. Terms act to reduce the claims prior to the application of covers. The subject net of terms is the de facto subject of covers. Covers are what actually produce a payout from the insuring party to the insured party.

Terms

Each term also has its own subject, equal to or smaller in scope than the contract's. Dimensions of scope are:

- Sub-schedule of risks
- Loss types
- Causes of loss

Loss types and causes of loss each are described as a tree in settings and taken as input to the calculation. For example, Building and Contents might each be a child loss type of PD, while WSSU and WSWI might both be child causes of loss of Windstorm.

The order in which terms apply is determined by their scope. Terms to disjoint sets of the subject claim can be processed in parallel, while terms whose scope is a subset of another term's go first and send their output to act as part of the term with larger scope's input. A tree (or trees) of terms is thus implied, which can be determined once per contract (not once per event).

All terms are considered together for purposes of ordering. If there is a tie, the term type can sometimes break a tie. All else being equal, minimum deductibles apply before maximum deductibles, which apply before sublimits. See Deductibles for more information about minimum and maximum deductibles.

Covers

Covers of a contract can also form (their own) tree (which, like the term tree, is also a one-time determination). The only mechanism for this is explicit. Covers can be on all of or on a subset of the contract's claims (net of terms). Or, covers can take (the sum or max of the payouts of) other covers as their subject. The contract payout is defined simply as the sum of all cover payouts from "top-level" covers. Top-level covers are those that are not the subject of some other cover in the contract. There can be no loops in the cover tree (A is on B, which is on A).

Read next: CDL Covers

CDL Covers

Covers have a share, payout function, and attachment. Each of these has its own amount. The order of operations is attachment, payout, share.

Attachment

Normally, loss above the attachment is eligible for the payout function. In other words, if the subject loss of the cover is S, and the attachment is A, max(S-A,0) is eligible to flow through the payout function and the share.

An option on the attachment is for it to be a franchise. In that case, it acts more like a trigger. If the subject loss is below the attachment, the cover produces 0 recoverable as usual. However, if the subject loss meets the attachment, the entire subject loss is eligible for the payout function.

Payout

Payout function is usually simply the limit function. If E is the eligible flow net of attachment, and L is the limit, then the payout P using a limit is simply min(E,L). The payout can also be a constant (Pay 10M xs 5M), which produces a constant payout in any event that satisfies the attachment.

Share

The share applies last and is simply a percentage that gets applied to the payout to form the final recoverable (R) of the cover.

Read next: CDL Terms

CDL Terms

Terms generally reduce claims. If you consider a claim as a vertical quantity, deductibles generally take the lower portion while sublimits take the higher portion. What remains in the middle continues on to the covers. The reduction from a deductible = D, and the reduction from a sublimit = X.

The input to a term is the loss from claims to the constrained subject of the term (S), as well as D and X of its "Child" terms in the tree. The output of a term is also an (S,D,X) vector, or (S.out, D.out, X.out).

Sublimits

To process sublimits, we must know the effect of the sublimit acting on the claim, as well as how that interacts with the incoming X of prior terms.

The effect of the sublimit acting on the claim is given as X = Max(S-D-A, 0) where S is the subject claim of the sublimit, D is input from prior terms, and A is the amount of the sublimit. Sublimits act net of deductible.

Deductibles

Deductibles can be somewhat more complicated than sublimits, but let's take the standard case first. The operation is similar to sublimits. We need to know the effect of the deductible acting on the claim, as well as how that interacts with prior terms.

The effect of the standard deductible acting on the claim is given as D = min(S,A). Franchise deductibles have a slightly different effect: If(S > A, 0, S).

"Max" deductibles have the opposite interaction from standard (or min) deductibles, ensuring the deduction is no more than a certain amount. For a max deductible, we take the minimum of the incoming D and the D produced by the deductible in isolation.

Net of Terms

Net of terms (R for recoverable) can be computed as R = S - D - X.

Read next: CDL Special Topics

CDL Special Topics

Amounts

Each term has an amount, as does each cover limit and each cover attachment. Amounts in CDL can be a monetary amount (100 USD) or an expression resolvable to a monetary amount. This resolution can be done once if it is not event-dependent (5% RCV Covered), or once per event if it is event-dependent (5% RCV Affected, 10% of Loss). The currency is the contract currency (as per contract declarations) unless otherwise overridden.

If the term applies "per risk" then the amount applies to each risk the term is "to." If the amount is a function that varies by risk, it must also be evaluated per risk, e.g. max(5k, 2% RCV Covered).

CDL enables the use of convenient multipliers: k, M, and B are shorthand for *10^3, *10^6, and *10^9.

Time Basis

Time basis of a term, cover limit, or cover attachment can be per occurrence (assumed if not stated) or "aggregate" across the period (period is the time between the declared inception and expiration).

Aggregate is dealt with via erosion. The amount used in the event reduces the effective amount that much for the next event.

Example: Aggregate deductible of 200. First event comes in and produces 150 of D at that term. Effective deductible for the next event in the period is 200-150 = 50. Attachments function similarly.

When considering erosion of aggregate cover elements, covers are at 100% with share applied as a final step - only to the payout.

Example: Cover has 50% of aggregate 10M limit. First event comes in at 8M. Payout is 50% * Min(8M,10M) = 4M. Amount of limit remaining for subsequent events is 10M-8M = 2M. The most that could be subsequently paid out is 50% of that amount = 1M.

Sections

Sections are an optional component of a contract that have a name and their own declarations and covers. Contract declarations are inherited by each section, but the section can append and override those with their own declarations. A section exposes its payout but not the payout of its individual covers. For order of operations, sections go before "contract-level" covers (covers outside any section).

Contract-level covers can be on:

- Other contract-level covers
- Sections

While section-level covers can be on:

- Other covers in the same section
- Other sections

"Covers by section" is a concise way to express in CDL that this contract's payout is the sum of section payouts. In such contracts, section-level covers cannot be on other sections.

Cover Subject

Like terms, covers can be unconstrained (apply to the full contract subject) or constrained in any of the three dimensions that constrain terms—subschedule, causes of loss, and loss types.

Other Inputs

In addition to claims, certain settings and subject exposure attributes are inputs to the calculation:

- Settings
 - Table of currency exchange rates
 - Hierarchy of causes of loss
 - Hierarchy of loss types
- Exposure Attributes
 - Which exposures (risk item and loss type) are part of which risk
 - The replacement cost value of each exposure
 - Which risks are part of which subschedule (if needed)

Detailed Contract Model Examples

The detailed examples in this section illustrate the calculations of the Risk Data Open Standard contract model. These examples assume claims from one flood event affecting one location and causing (150k, 30k, 20k) USD loss to (Building, Contents, BI) unless stated otherwise.

Examples include:

- Simple Covers and Terms
 - Simple Covers
 - Simple Sublimits
 - Simple Deductibles
- Simple Sections
- Complex Terms
 - Max Deductible
 - Percent of Loss Deductible
 - Percent of RCV Covered
 - Percent of RCV Affected
 - Franchise Deductible
 - Per Risk Terms
 - Absorbing a Deductible
 - Aggregate Terms
- Complex Covers
 - Aggregate Limits and Attachments
 - Cover Trees
 - Constrained Covers
 - Payout Functions
 - Step Policies

RMS[®] CONFIDENTIAL Page 10

- Franchise Attachment
- Complex Covers

Simple Covers and Terms

These examples assume claims from one flood event affecting one location and causing (150k, 30k, 20k) USD loss to (Building, Contents, BI) unless stated otherwise.

Simple Covers

A contract in Contract Definition Language (CDL) must have at least one cover. A cover must have a share, can optionally have a limit, and can optionally have an attachment. For a simple contract with just one cover, the cover acts on the sum of claims in the contract subject. Contract Subject refers to the position to which the contract applies (whether a risk, a schedule of risks, or a position in a structure).

- 1. First the attachment applies, if one is present.
 - If the cover's subject is less than the attachment, the cover pays zero.
 - If the cover's subject is more than the attachment, the subject minus the attachment is sent to the limit.
 - If there is no attachment, "net of attachment" is the cover's subject loss.
- 2. The limit applies, if one is present.
 - If the net of attachment is bigger than the limit, the full limit is sent on to the share.
 - If the net of attachment is positive but smaller than the limit, the net of attachment is sent on to the share.
 - If there is no limit, the "limited loss" is the net of attachment.
- 3. The share applies
 - The limited loss is multiplied by the share to produce the contract payout.

To see how CDL allows for more complexity in covers, see Complex Covers. For this simple cover example, assume the following CDL:

```
Contract
Declarations
Currency is USD
Covers
80% share of 100k xs 20k
```

A claim of 50k flows through the cover:

- 1. Net of attachment is 50k-20k = 30k
- 2. 30k is less than 100k, so limited loss is 30k.
- 3. 80% of 30k is **24k**, which is the contract payout.

Similarly:

- A claim of 10k would produce 0 payout.
- A claim of 20m would produce a payout of 80k.

Simple Sublimits

In addition to covers, CDL contracts can have *terms* that act first on the subject loss to reduce it before sending it to the covers for payout. The two types of terms are *sublimits* and *deductibles*. Sublimits act to cap subsets of the subject claims. Any loss in excess of sublimits is not eligible for payout by covers.

- If the claim is below the sublimit, the total claim is eligible for recovery.
- If the claim is above the sublimit, only as much as the sublimit is eligible for recovery.

Assume that a sublimit of 60k for Building, Contents is added to the previous Simple Covers example:

```
Contract
Declarations
Currency is USD
Covers
100% share of 100k
Sublimits
60k for Building, Contents
```

The 60k sublimit applies to the 180k of claims from Building and Contents. Loss above that is not eligible for recovery. The "net of terms" to which the cover applies is 60k + 20k BI = 80k. Since 80k is less than the cover limit of 100k, the contract payout is **80k**.

Simple Deductibles

Like sublimits, deductibles are contract terms. They act to reduce the subject claims and send that which is recoverable to the covers. Unlike sublimits, which discard losses above a certain threshold, deductibles discard losses below a certain threshold.

- If the claim is below the deductible, none of the claim is recoverable.
- If the claim is above the deductible, the claim minus the deductible is potentially recoverable.

Assume that a deductible of 20k for Building, Contents is added to the previous Simple Covers example:

```
Contract
Declarations
Currency is USD
Covers
100% share of 100k
Deductibles
20k for Building, Contents
```

Of the 180k of claims for building + contents, 20k is deducted. 160k is potentially recoverable. This is sent along with the 20k of BI claims into the cover, which still applies the 100k limit.

Simple Sections

These examples assume claims from one flood event affecting one location and causing (150k, 30k, 20k) USD loss to (Building, Contents, BI) unless stated otherwise.

Sections are an optional component of CDL contracts. Sections have cover(s) and optionally declarations that apply only to the section. Covers of the contract can be "By Section" as a shorthand for the contract to pay the sum of the sections. Otherwise, the contract can have covers that are specifically "on" sections and process them further.

```
Contract

Declarations

Currency is USD

Covers by Section

Sections

Section Policy1

Declarations

LOB is Primary

Covers

100% Share of 50k

Section Policy2

Declarations

LOB is Excess

Covers

100% Share of 100k xs 50k
```

This contract pays 50k for policy1 and 100k for policy2, for a total of 150k.

Complex Terms

Contract Definition Language (CDL) includes features to model complex terms.

These examples assume claims from one flood event affecting one location and causing (150k, 30k, 20k) USD loss to (Building, Contents, BI) unless stated otherwise.

Max Deductible

Sometimes when there are levels of deductibles, the higher level deductible can act to ensure that the deduction taken is **no more than** a certain amount. This is often called a *maximum deductible* or *max deductible*.

```
Contract
Declarations
Currency is USD
Covers
100% share
Deductibles
30k max
30k for Building
30k for Contents
30k for BI
```

Assuming the same claims as our previous examples (150k, 30k, 20k), the three lower-level deductibles produce deductions of 30k, 30k, 20k. Since there is a maximum deductible of 30k, 30k rather than 80k is deducted. The contract payout is 200k-30k = **170k**.

Percent of Loss Deductible

A percent of loss deductible deducts a percentage of the loss for its exposure instead of a fixed amount. In the following example, claims are 150k + 30k + 20k = 200k. A 10% deductible deducts 20k, which results in a payout of 180k in this case.

```
Contract
Declarations
Currency is USD
Covers
100% share
Deductibles
10% of Loss
```

Percent of RCV Covered

The exposure in these examples has a total insurable value (TIV) of 1M for Building, 100k for Contents, and 50k for BI. A 2% RCV Covered deductible on this location = 23k. If a second location with identical values is covered by the policy and takes no loss, a percentage of RCV Covered deductible takes 2 * 23k = 46k.

```
Contract
Declarations
Currency is USD
Covers
100% share
Deductibles
2% RCV Covered to R1, R2
```

Even though R2 takes no loss, the deductible of 46k is taken from the 200k claim of R1, and the contract pays 154k.

Percent of RCV Affected

In contrast to RCV Covered, RCV Affected takes as its basis only exposure with claims. Since only one of the two 1.15M locations has a claim, the deductible is 2% * 1.15M = 23k, and the payout is 177k.

```
Contract
Declarations
Currency is USD
Covers
100% share
Deductibles
2% RCV Affected to R1, R2
```

Franchise Deductible

A standard deductible deducts the full claim if the claim is less than or equal to the deductible. It deducts the amount of the deductible from the claim if the claim is larger than the deductible.

A franchise deductible deducts the full claim if the claim is less than or equal to the deductible, but deducts zero from the claim if the claim is larger than the deductible.

In the following examples, the BI claim is 20k.

```
Contract
Declarations
Currency is USD
Covers
100% share
```

```
Deductibles
10k Franchise for BI
```

Since the 20k BI claim is larger than 10k, the BI claim is eligible to be paid in full. With no other covers or terms, this contract pays the full **200k**.

```
Contract
Declarations
Currency is USD
Covers
100% share
Deductibles
30k Franchise for BI
```

Since the 20k BI claim is smaller than 30k, the BI claim is not eligible to be paid. With no other covers or terms, this contract pays the full **180k**.

Per Risk Terms

Terms can apply Per Risk, which means rather than applying once to the total of the claims of their exposure, there is one term per risk.

```
Deductibles
10k to R1, R2 per risk
is equivalent to:
```

```
Deductibles
10k to R1
10k to R2
```

Absorbing a Deductible

A common practice in insurance is to allow losses over a sublimit to act as credit to a deductible. Deductibles in the RDOS contract model work this way by default.

```
Contract
Declarations
Currency is USD
Covers
100% share
Sublimits
5k for Contents
Deductibles
10k
```

RMS[®] CONFIDENTIAL

The Contents sublimit makes 25k of claims ineligible for recovery. The 10k deductible is satisfied and does not deduct an additional 10k. The payout is **175k** in this case.

Aggregate Terms

Sublimits and deductibles can apply to the claims of one event or in the aggregate to the claims of all events during the contract's lifetime. Processing is done in chronological order, with the "remaining" amount of the term not yet used after each event as the effective amount for subsequent events.

The following example assumes three events in a period, rather than a single event. In each event, a different location is impacted with claims of 150k, 30k, 20k to Building, Contents, BI.

Contract
Declarations
Currency is USD
Covers
100% share
Sublimits
300k Aggregate

The first event pays in full but uses 200k of the 300k sublimit. The effective sublimit before the second event is 100k. The 100k sublimit applies to the claim of 200k from the second event, which pays out 100k but reduces the effective sublimit to 0 for the third event. The third event does not trigger a payout.

Complex Covers

Contract Definition Language (CDL) includes features to model complex covers.

These examples assume claims from one flood event affecting one location and causing (150k, 30k, 20k) USD loss to (Building, Contents, BI) unless stated otherwise.

Aggregate Limits and Attachments

Like terms, each individual amount in each cover can act cumulatively over time. The following example assumes a cover with both attachment and limit aggregate and three events in the contract period, similar to Aggregate Terms

```
Contract
Declarations
Currency is USD
Covers
100% Share of 250k aggregate xs 250k aggregate
```

- 1. In the first 200k event, the attachment is not met but is eroded down to 50k for subsequent events.
- 2. In the second 200k event, the remaining 50k of attachment is used, and the layer pays 150k while eroding the limit for subsequent events to 100k.
- 3. In the third 200k event, there is no effective attachment, and the remaining 100k of limit is used.

Cover Trees

Covers are **not** automatically put in order but can be **explicitly** constructed such that the payout of one becomes the subject of the other. When a contract does not have just one top-level cover, it pays the sum of its top-level covers (the ones no other Cover takes as their Subject).

Mechanically, in addition to being directly on the net of terms, a cover can be applied **on** another cover or the sum of a list of covers. A common use case for this is an occurrence layer capped by an aggregate limit.

```
Contract
Declarations
Currency is USD
Covers
Occ: 100% Share of 125k xs 50k
Agg: 100% Share of 200k aggregate ON occ
```

Event	Claim	Payout
1	200,000	125,000
2	200,000	75,000
3	200,000	0

Firstly, note *Agg* and *Occ* are just labels to identify the covers. No semantics come from the names themselves. Each event of 200k goes first through the Occ layer, then sends the result of that calculation (125k) through the Agg layer. The final contract payout is the sum of top-level covers (those covers on which there are no other covers) which in this example is the one cover, Agg. If there were any terms, the Occ cover would apply to the net of terms and then the Agg cover would apply to the output of the Occ cover.

Constrained Covers

Like terms, covers can also have subject constraints *from* a list of loss types, *to* a list of risks, *by* a list of causes.

The following example illustrates a per-risk location deductible and covers to sets of risks.

```
Contract
Declarations
Currency is USD
Covers
Group1: 100% share of 300k xs 100k to R1, R2
Group2: 100% share of 400k xs 100k to R3
Deductibles
25k per risk
```

Replicating the R1 claims (150k, 30k, 20k) to R2 and R3, the payout net of terms to Group1 of 350k and to Group2 of 175k produce layer payouts of 250k and 75k respectively. The contract pays **325k**.

Payout Functions

Limit is the most common payout function for a cover. Given a net of attachment, the limit function is max(net of attachment, 0). Other payout functions are sometimes used in practice and are supported in CDL. The following example illustrates another common payout function to pay a constant on the attachment being exceeded.

```
Contract
Declarations
Currency is USD
Covers
100% share of Pay 300k xs 100k
```

Since the subject loss is 200k, the attachment is met and the cover pays 300k.

More complex payout functions can also be used, as required.

```
Contract
Declarations
Currency is USD
Covers
100% share of Pay Max (1M - Subject, 0)
```

The contract pays 800k in this case, since max(1M-200k,0) = 800k.

Step Policies

Simple Step Policy

Taiwan earthquake policies are examples of simple step policies. If the claim is bigger than S1, pay P1, else pay 0.

```
Contract
Declarations
Currency is USD
Covers
100% share of Pay 1M xs 150k
```

The 200k claim is bigger than 150k, so the contract pays 1M.

Medium-Complexity Step Policies

In Japan, step policies are written with multiple steps.

- 1. If claim > S1, pay P1
- 2. If S1 > claim > S2, pay P2
- 3. And so on

CDL accomplishes this with payout covers per step, with different attachments, and a cover *on* those steps that pays the max of any step (so that the contract does not accidentally pay the sum of all steps).

The following example illustrates that, in addition to operating on the sum of a list of covers, a cover can be set to act on the *max* or *min* of a list of covers, or even on the *diff* of two covers.

```
Contract

Declarations

Currency is USD

Covers
```

RMS® CONFIDENTIAL

```
step1: 100% share of Pay 50k xs 75k
step2: 100% share of Pay 150k xs 150k
step3: 100% share of Pay 250k xs 225k
final: 100% share on max(step1, step2, step3)
```

A subject loss of 200k will hit *step2* and pay 150k. *Step1* also pays its 50k. The *final* cover ensures that only the **150k** is paid in the end.

Complex Step Policy

Steps might be for Building *or* Contents (but not the sum of Building *and* Contents). Sometimes step payouts are limited to the subject claim. See Expressions for more information about the mechanics used in this example.

```
Contract
Declarations
Currency is USD
Covers

step1B: 100% share of Pay min (Subject, 50k) xs 75k for Building step1C: 100% share of Pay min (Subject, 50k) xs 75k for Contents step1: 100% share on Max(step1B, step1C)

step2B: 100% share of Pay min(Subject, 150k) xs 150k for Building step2C: 100% share of Pay min(Subject, 150k) xs 150k for Contents step2: 100% share on Max(step2B, step2C)

step3B: 100% share on Max(step2B, step2C)

step3B: 100% share of Pay min(Subject, 250k) xs 225k for Building step3C: 100% share of Pay min(Subject, 250k) xs 225k for Contents step3: 100% share on Max(step3B, step3C)

final: 100% share on max(step1, step2, step3)
```

Step1B pays 50k, and Step1C pays 0, so Step1 pays 50k all for Building. Step2 and Step3 pay 0 as the attachments are not met. Final pays 50k, all for Building.

Franchise Attachment

Like deductibles, attachments can be *franchise*. The covers limit acts on the cover's entire subject if the subject > attachment, instead of its subject net of attachment.

```
Contract

Declarations

Currency is USD

Covers

step1: 100% share of 300k xs 125k franchise
```

Since the 200k claim is larger than 125k, the full claim is eligible to be paid through the limit, and the contract pays 200k.

Special Topics

The content of this section addresses the following special topics:

- Expressions
 - Simple Expressions
 - Dynamic Expressions
- Currency
- Subschedules
- Sections
 - Cashflows
 - Reinstatements

Expressions

Expressions in CDL can be simple expressions or dynamic expressions.

Simple Expressions

Amounts in CDL can be expressions in addition to being numbers, such as when the share is a rational fraction without a terminating decimal representation, like 1/3.

```
Contract
Declarations
Currency is USD
Covers
1/3 share of 100k
```

The contract pays 33,333.33.

Expressions are also used when the amounts are functions of other amounts.

```
Contract
Declarations
Currency is USD
Covers
100% share
Sublimits
60k for Building
60k * 0.2 for Contents
60k * 0.1 for BI
```

The contract pays **78k**. All the sublimits apply, and the claims are limited to 60k + 12k + 6k.

Nestable arithmetic and logical expressions are allowed.

```
Contract

Declarations

Currency is USD

Covers

100% share of ((20k * 5) / max((5-3),1)) - 10k
```

The contract pays 40k, which is how the expression for the limit is calculated.

RMS[®] CONFIDENTIAL Page 26

Dynamic Expressions

Some words in CDL evaluate to a value (either per analysis or per event), which can be used to evaluate the expression of the amount.

Replacement Cost Value (RCV)

```
Contract

Declarations

Currency is USD

Covers

100% Share

Deductibles

Max(2% RCV Covered, 25k) per risk
```

The following example duplicates the (150k, 30k, 20k) R1 claims to R2, but doubles the exposure values of R2.

R1 has a TIV of 1.15M (making 2% RCV Covered = 23k).

R2 has a TIV of 2.3M (making 2% RCV covered = 46k).

The contract pays 329k.

- 1. R1 total payout is 175k on claim of 200k.
 - Deductible expression Max(2% RCV Covered, 25k) evaluates to 25k when RCV covered = 1.15M.
- 2. R2 total payout is 154k on claim of 200k.
 - Deductible expression Max(2% RCV Covered, 25k) evaluates to 46k when RCV covered = 2.3M

Subject

The share of a layer can be a function of loss. A share could be 0% at the attachment point, max participation at the limit, and a linear function in between. The keyword *Subject* in the expression indicates the cover's subject loss.

```
Contract
Declarations
Currency is USD
Covers
Min(-1/2 + Subject / 300k, 1/2) Share of 150k xs 150k
```

In this contract, the maximum participation is only 50%. The share evaluates to 0 at the attachment (subject=150k), 50% at the limit (subject=300k), and interpolates for any subject loss in between. Since the subject of this cover is 200k, the expression evaluates to 1/6. 200k in the layer of 150k xs 150k is 50k. So the payout is 1/6 of 50k. So the payout is 1/6 of 50k.

Currency

A contract in CDL declares a currency that applies to all contract amounts unless otherwise specified. Any amount in CDL can be in any currency. Currency symbols and their exchange rates are read from settings, which are user-definable.

The following example assumes an exchange rate of 1 USD = 8 HKD. Assume the analysis remains in USD, and claims remain in USD, but the contract is in HKD. Then, a 100k limit (in HKD) is 100k / 8 = 12.5k USD limit.

```
Contract
Declarations
Currency is HKD
Covers
100% share of 100k
```

The contract requires a limit of 800k (HKD) in order to get the same results as with a 100k USD limit.

```
Contract
Declarations
Currency is HKD
Covers
100% share of 800k
```

It is rare but possible for a contract to have amounts in different currencies. The following example illustrates a 200k USD claim --(150k, 30k, 20k) to (Building, Contents, BI). For a 150k USD attachment, there would be a 50k USD payout (which is less than the 100k USD limit). To illustrate the foreign currency attachment, assume a 150k * 7 = 1050k RMB attachment.

```
Contract
Declarations
Currency is HKD
Covers
100% share of 800k xs 1050k RMB
```

The following example illustrates further with JPY as 110:1 with USD and EUR as 0.9:1 with USD (and HKD and RMB are still 8:1 and 7:1, respectively). The contract pays **45k**.

```
Contract
Declarations
Currency is HKD
Covers
100% share of 800k xs 105k RMB
Sublimits
5500k JPY for Building, Contents
```

Deductibles
9k EUR for BI

Subschedules

Each Cover and Term can be to a Risk or to a list of Risks. To keep the main body of CDL concise and readable, a list of risks can be defined as a named subschedule. Then the cover or term applies to the subschedule.

```
Contract
Declarations
Currency is USD
Covers
100% share
Sublimits
300k to S1
100k to R3
Deductibles
25k to S1 per risk
Subschedules
S1: R1, R2
```

This example duplicates the R1 claims from the event to R2 and R3, so they each have (150k, 30k, 20k) of loss in this event. The payout is **400k** since the deductible is taken from the 400k of the claims of R1 and R2 combined, which is then sent through the 300k sublimit while the 200k claim of R3 is sublimited to 100k.

In RDOS, subschedules are typically expressed in an external table, but can still be referenced in the "to" clause of contract terms and covers.

Sections

In addition to the main contract payout, it is often useful to have the contract output other results. Sections can be used for this. The contract in Simple Sections outputs not only its loss, but each section's loss. For example, a facultative cession could cover just the *Policy1* section of the contract. Contract sections become positions and can be referenced in a structure using *ContractName#SectionName*.

Sections can also be used to represent and calculate cashflows of a contract other than the cashflow that turns the claim into a gross payout. Common examples are premium, reinstatement premium, and taxes. CDL provides support for cashflows and reinstatement premiums, and creates the sections that can be referenced in structures (e.g., Contract1#Premium).

Cashflows

Cashflows can be added to a contract and generate section payouts (on the contract inception date by default). In the previous sections of this document, all contract payouts were reactive to events that came from their subject's claims. A cashflow can be a constant or a percentage of another cashflow.

```
Contract
Declarations
Currency is USD
Inception is 1 Jan 2019
Expiration is 31 Dec 2019
Cashflows
Premium is 1M
Brokerage is 7% of Premium
Covers
100% Share of 60k xs 20k
```

Reinstatements

Reinstatements can be added to a contract in order to represent and calculate reinstatement premium and output it as its own position. Given a premium as a cashflow and a cover with an occurence limit, the reinstatement generates a charge of some percentage of that premium when the occurence limit is used. This percentage can vary by event, and each reinstatement can also optionally be proportional as to the amount of the limit used and/or the amount of time remaining in the contract.

Simple example:

```
Contract
Declarations
Currency is USD
Inception is 1 Jan 2019
Expiration is 31 Dec 2019
Cashflows
```

```
Premium is 10k
Reinstatements
RP: 1 @ 100% PRC on Occ with Premium
Covers
Occ: 100% Share of 60k xs 20k
Agg: 100% Share of 120k Aggregate on Occ
```

The claim of 200k would pay 60k while the reinstatement would pay 100% of the premium times the percent of the limit used (100%), which is 10k.